
1   log using "20_data_process_log.txt", text replace
2   //// RCT Survey: Data Processing
3   //// Lucas Reddinger <jlr@lucasreddinger.com>, David Levine, Gary Charness
4   //// 2021 April 24
5   
6   cls
7   clear
8   set more off
9   

10   cd "E:\lucas\projects_private_git\academia\projects\vaccine\rct"
11   
12   ************************
13   **** Setup Prolific data
14   ************************
15   
16   ** RCT 1 FINAL PROLIFIC SURVEY DATA
17   import delimited "data\data_prolific_survey_rct1_final.csv", varnames(1) rowrange(4)
18   destring _all, replace
19   
20   rename participant_id prolific_pid
21   
22   // We should not have any duplicates.
23   sort prolific_pid session_id
24   quietly by prolific_pid (session_id): gen dup_pid = cond(_N==1,0,_n)
25   tab dup_pid
26   
27   keep prolific_pid status age countryofbirth currentcountryofresidence ///
28   employmentstatus firstlanguage nationality sex studentstatus
29   
30   save "data\data_prolific_survey_rct1_final.dta", replace
31   clear
32   
33   ** RCT 2 RECENT PROLIFIC SURVEY DATA
34   import delimited "data\data_prolific_survey_rct2_2021-04-23_1134.csv", varnames(1) rowrange(4)
35   destring _all, replace
36   
37   rename participant_id prolific_pid
38   
39   // We should not have any duplicates.
40   sort prolific_pid session_id
41   quietly by prolific_pid (session_id): gen dup_pid = cond(_N==1,0,_n)
42   tab dup_pid
43   
44   keep prolific_pid status age countryofbirth currentcountryofresidence ///
45   employmentstatus firstlanguage nationality sex studentstatus
46   
47   save "data\data_prolific_survey_rct2_2021-04-23_1134.dta", replace
48   clear
49   
50   *********************************
51   **** Import Qualtrics survey data
52   *********************************
53   
54   import delimited "data\data_qualtrics_survey_2021-04-23_0939.csv", varnames(1) rowrange(4)
55   destring _all, replace
56   
57   *********************
58   **** SHUFFLE OUTCOMES
59   *********************
60   // Obviously this will be removed when we are ready to analyze the data
61   
62   set seed 578588
63   gen double shuffle1 = runiform()
64   gen double shuffle2 = runiform()
65   
66   sort shuffle1
67   gen long which = _n
68   sort shuffle2
69   gen int manip_chk_new = manip_chk[which], after(manip_chk)
70   gen int vax_likely_new = vax_likely[which], after(vax_likely)
71   gen int vax_delay_new = vax_delay[which], after(vax_delay)
72   gen int vax_child_new = vax_child[which], after(vax_child)
73   
74   list which vax_likely* in 1/30
75   
76   drop manip_chk vax_likely vax_delay vax_child shuffle* which
77   rename manip_chk_new manip_chk
78   rename vax_likely_new vax_likely

79   rename vax_delay_new vax_delay
80   rename vax_child_new vax_child
81   
82   **************************
83   **** Qualtrics data basics
84   **************************
85   
86   **** Drop variables that have no information for us
87   drop startdate enddate responseid status progress recipient*
88   drop externalreference distributionchannel
89   
90   **** These click data are not very interesting to us
91   drop *_firstclick *_lastclick *_clickcount
92   
93   **** Drop some other junk
94   drop userlanguage religious*
95   
96   **** Convert date-time to stata clock
97   gen double recordedclock = clock(recordeddate, "YMD hms"), after(recordeddate)
98   format recordedclock %tc
99   drop recordeddate

100   
101   **** Recode race to black
102   gen d_black = 0, after(d_race)
103   replace d_black = 1 if d_race==2 | d_race==12
104   replace d_black = . if d_race>=.
105   drop d_race
106   
107   **** Recode natural indicators
108   replace d_latinx = 0 if d_latinx == 2
109   replace d_parent = 0 if d_parent == 2
110   
111   **** Rename d_smalltown_rural
112   rename d_s d_smalltown_rural
113   
114   ************************
115   **** Merge Prolific data
116   ************************
117   
118   ** RCT 1 FINAL PROLIFIC SURVEY DATA
119   merge m:1 prolific_pid using "data\data_prolific_survey_rct1_final.dta", update
120   drop _merge
121   
122   ** RCT 2 RECENT PROLIFIC SURVEY DATA
123   merge m:1 prolific_pid using "data\data_prolific_survey_rct2_2021-04-23_1134.dta", update
124   drop _merge
125   
126   **************************
127   **** Label variable values
128   **************************
129   
130   #delimit ;
131   
132   label variable manip_chk "How many people have died?";
133   
134   label values manip_chk manip_chk_label;
135   
136   label define manip_chk_label
137   2 "Over 200,000"
138   3 "Over 300,000"
139   4 "Over 400,000"
140   5 "Over 500,000"
141   6 "Over 600,000";
142   
143   label variable vax_likely "How likely get vax described above?";
144   
145   label values vax_likely vax_likely_label;
146   
147   label define vax_likely_label
148   1 "Highly likely"
149   2 "Somewhat likely"
150   3 "Slightly likely"
151   4 "Neither likely nor unlikely"
152   5 "Slightly unlikely"
153   6 "Somewhat unlikely"
154   7 "Highly unlikely"
155   8 "DK / prefer not to say";
156   

157   label variable vax_delay "When would get vax described above?";
158   
159   label values vax_delay vax_delay_label;
160   
161   label define vax_delay_label
162   1 "Already have taken a vaccine dose"
163   2 "As soon as possible"
164   3 "When more people have been vaccinated"
165   4 "Unsure when I will take the vaccine"
166   5 "Unlikely to ever take the vaccine"
167   6 "Definitely will not take the vaccine"
168   7 "DK / prefer not to say";
169   
170   label variable covid_hist "Your experience w COVID-19 virus?";
171   
172   label values covid_hist covid_hist_label;
173   
174   label define covid_hist_label
175   1 "Positive, overnight hospital stay"
176   2 "Positive, no overnight hospital stay"
177   3 "Strongly suspect was positive, not tested"
178   4 "Strongly suspect was positive, tested negative"
179   5 "Believe didn't have"
180   6 "DK / prefer not to say";
181   
182   label variable vax_child "How likely get child vax as above?";
183   
184   label values vax_child vax_likely_label;
185   
186   label variable d_reside "Residence";
187   
188   label values d_reside d_reside_label;
189   
190   label define d_reside_label
191   1 "US"
192   2 "Other";
193   
194   label variable d_age "Age";
195   
196   label values d_age d_age_label;
197   
198   label define d_age_label
199   1 "Under 18"
200   2 "18 - 24"
201   3 "25 - 34"
202   4 "35 - 44"
203   5 "45 - 54"
204   6 "55 - 64"
205   7 "65 - 74"
206   8 "75 - 84"
207   9 "85 or older";
208   
209   label variable d_parent "Children";
210   
211   label values d_parent d_parent_label;
212   
213   label define d_parent_label
214   0 "No"
215   1 "Yes";
216   
217   label variable d_black "Black?";
218   
219   label values d_black d_black_label;
220   
221   label define d_black_label
222   0 "No"
223   1 "Yes";
224   
225   label variable d_latinx "Hispanic or Latinx?";
226   
227   label values d_latinx d_latinx_label;
228   
229   label define d_latinx_label
230   0 "No"
231   1 "Yes";
232   
233   label variable d_gender "Gender?";
234   

235   label values d_gender d_gender_label;
236   
237   label define d_gender_label
238   1 "Male"
239   2 "Female"
240   3 "Other";
241   
242   label variable d_vote "Vote in 2020?";
243   
244   label values d_vote d_vote_label;
245   
246   label define d_vote_label
247   1 "Biden"
248   2 "Trump"
249   3 "Other/Didn't vote/DK";
250   
251   label variable d_pols "Political scale?";
252   
253   label values d_pols d_pols_label;
254   
255   label define d_pols_label
256   1 "Extremely lib"
257   2 "Liberal"
258   3 "Slightly lib"
259   4 "Moderate"
260   5 "Slightly cons"
261   6 "Conservative"
262   7 "Extremely cons"
263   8 "Don't know";
264   
265   label variable d_relig "Religious preference?";
266   
267   label values d_relig d_relig_label;
268   
269   label define d_relig_label
270   1 "Pr - Evangelical"
271   2 "Pr - Mainline"
272   3 "Catholic"
273   4 "Mormon"
274   5 "Other Chr"
275   6 "Jewish"
276   7 "Muslim"
277   8 "Eastern"
278   9 "None"
279   10 "Other / DK";
280   
281   label variable d_relobs "How often observe religion?";
282   
283   label values d_relobs d_relobs_label;
284   
285   label define d_relobs_label
286   1 "Sev/wk"
287   2 "Once/wk"
288   3 "Sev/mo"
289   4 "Once/mo"
290   5 "Sev/yr"
291   6 "Once/yr"
292   7 "LT once/yr"
293   8 "Never"
294   9 "DK / NA";
295   
296   label variable d_popdens "Small town or rural?";
297   
298   label values d_popdens d_popdens_label;
299   
300   label define d_popdens_label
301   1 "City or suburb"
302   2 "Large town"
303   3 "Small town"
304   4 "Rural area";
305   
306   #delimit cr
307   
308   **************
309   **** Save data
310   **************
311   
312   save "data\data_shuffled_2021-04-23_0939.dta", replace

313   
314   *************
315   **** Codebook
316   *************
317   
318   codebook
319   
320   **************
321   **** Close log
322   **************
323   
324   log close
325   


